

R. Brigola, TH Nürnberg Georg Simon Ohm, 2013

Mathematica - Notebooks als Bonusmaterial zum Lehrbuch

**[1] Rolf Brigola Fourier-Analysis und Distributionen,
Eine Einführung mit Anwendungen,
edition swk, Hamburg 2013**

Teil 1 Demonstrationsbeispiele zu Fourierreihen mit *Mathematica*

Referenzen zu Kapiteln, Abschnitten, Seitenzahlen beziehen sich im Folgenden meist auf das genannte Lehrbuch. Einige wenige weitere Referenzen findet man am Ende des Notebooks.

Ziel dieses ersten Notebooks zum Thema ist, Nutzern zunächst einige Grundlagen über Fourierreihen mit Hilfe von *Mathematica* vorzustellen bzw. in Erinnerung zu bringen. Dabei werden einige einfache *Mathematica*-Befehle verwendet, die man zur Berechnung von Fourierentwicklungen und zur Erzeugung von Grafiken für eine Veranschaulichung der Sachverhalte benutzen kann. Das Notebook ist mit *Mathematica* 9 unter Windows 7 erstellt. Dieses und weitere Demo-Notebooks zum Thema findet man unter folgender URL des Autors: www.stiftung-swk.de/mathematica

In nachfolgenden Notebooks sollen dann einige typische Anwendungsbeispiele gezeigt und der Nutzen der Fourier-Analysis für Approximationsaufgaben und Signalverarbeitung demonstriert werden. Dort betrachten wir dann insbesondere die DFT, DCT und Tschebyscheff-Polynome (oft auch in der Schreibweise Chebychev-Polynome zu finden) und einige ihrer typischen Eigenschaften, die für Anwendungen relevant sind.

Anmerkung des Autors: Ich bin kein Experte für die schier unerschöpflichen Möglichkeiten, die ein wirklich geübter, vertiefter Umgang mit *Mathematica* bietet. Ich habe mich daher im Wesentlichen bemüht, den Stoff mit diesem Angebot zu veranschaulichen und zu zeigen, wie man mit (zum Teil wahrscheinlich oft zu umständlichen, dafür aber auch für *Mathematica*-Anfänger wie mich selbst transparenten) Anweisungen die behandelten Inhalte erschließen kann.

Beispiele zu Fourierreihen – Entwicklungen

Eine fundamentale Rolle beim Studium von Fourierreihen spielen Dirichlet-Kerne und Fejér-Kerne. Partialsummen von Fourierreihen periodischer Funktionen sind Faltungen mit Dirichlet-Kernen (vgl. [1], Kapitel 3 und Kapitel 6).

1. Dirichlet - Kerne

Wir betrachten den 2π - periodischen Dirichlet-Kern $\text{Dir}[t, N]$ vom Grad $N \geq 1$:

```
In[1]:= Dir[t_, N_] = 1 + 2 Sum[Cos[k t], {k, 1, N}]
```

```
Out[1]= 1 + 2 Cos[1/2 (1 + N) t] Csc[t/2] Sin[N t/2]
```

Nachfolgend in einer trigonometrischen Form


```
In[2]:= TrigToExp[1 + 2 Cos[1/2 (1 + N) t] Csc[t/2] Sin[N t/2]]
```

```
Out[2]= 1 + ((e^{-1/2 i N t} - e^{i N t/2}) (e^{-1/2 i (1+N) t} + e^{1/2 i (1+N) t})) / (e^{-i t/2} - e^{i t/2})
```

Als Grafik die Dirichlet-Kerne vom Grad 3 und 6.
Sie oszillieren mit wachsenden Graden N immer mehr.

Für $N \rightarrow \infty$ existiert an keiner Stelle ein Grenzwert.

```
In[3]:= g1 := Plot[Dir[t, 3], {t, -2 Pi, 2 Pi}, PlotStyle -> Directive[
  Red, Thickness[0.005]], PlotRange -> All]
g2 := Plot[Dir[t, 6], {t, -2 Pi, 2 Pi}, PlotStyle -> Directive[
  Blue, Thickness[0.005]], PlotRange -> All]
Show[g1, g2]
```



Wir beobachten, dass ein Dirichlet-Kern vom Grad N die mögliche Maximalzahl von $2N$ Nullstellen pro Periode besitzt. Auch Dirichlet-Kerne anderer Periodendauer besitzen jeweils $2N$ Nullstellen pro Periode.

Die Mittelwerte der Dirichlet-Kerne über eine Periode sind Eins. Hier als Beispiel:

```
In[6]:= Integrate[Dir[t, 5], {t, 0, 2 Pi}] / (2 Pi)
```

```
Out[6]= 1
```


2. Die Fourierreihen - Entwicklung des Sägezahns

Als erste periodische Funktion betrachten wir den 2π -periodischen Sägezahn im "Zeitausschnitt" von -4π bis 2π . Zunächst der Ausschnitt in $[0, 2\pi]$

```
In[7]:= Saegezahn[t_] = (Pi - t) / 2 (UnitStep[t] - UnitStep[t - 2 Pi])
```

```
Out[7]=  $\frac{1}{2} (\pi - t) (\text{UnitStep}[t] - \text{UnitStep}[-2\pi + t])$ 
```

```
In[8]:= Plot[Sum[Saegezahn[t - k 2 Pi], {k, -4, 2}], {t, -4 Pi, 2 Pi},
  PlotStyle -> Directive[RGBColor[0.127, 0.121, 0.36], Thickness[0.005]]]
```



Nun zur Fourierreihen-Entwicklung des Sägezahns. Mathematica bietet dafür vorhandene Prozeduren an und verwendet dazu als Standard 2π -periodische Funktionen, definiert im Intervall $[-\pi, \pi]$. Wir definieren daher als Funktion f den entsprechenden Ausschnitt des Sägezahns.


Das Kommando `FourierTrigSeries` liefert die Partialsumme der Fourierreihen-Entwicklung mit der angegebenen Anzahl von Oberschwingungen.

Das **Gibbs-Phänomen** an den Sprungstellen des Sägezahns ist deutlich zu erkennen.

```
In[9]:= f[t_] := Saegezahn[t + 2 Pi] + Saegezahn[t]
  FRsaegezahn[t_] = FourierTrigSeries[f[t], t, 7]
```

```
Out[10]=  $\sin[t] + \frac{1}{2} \sin[2t] + \frac{1}{3} \sin[3t] + \frac{1}{4} \sin[4t] + \frac{1}{5} \sin[5t] + \frac{1}{6} \sin[6t] + \frac{1}{7} \sin[7t]$ 
```

```
In[11]:= Plot[FRsaegezahn[t], {t, -2 Pi, 2 Pi},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
```



Der Schlüssel zum Verständnis von Fourierreihen-Entwicklungen:

Faltungen mit Dirichlet-Kernen, später mit Fejér-Kernen
und anderen Summationskernen (vgl. [1], S. 23 und S. 103 - 110)

Man erhält die gleiche Partialsumme aus der 2π -periodischen Faltung
des Sägezahns mit dem 2π -periodischen Dirichlet-Kern vom Grad 7 :

In[12]:= `Faltung[t_] = 1 / (2 Pi) Integrate[Saegezahn[s] * Dir[t - s, 7], {s, 0, 2 Pi}]`

Out[12]=
$$\frac{1}{210} (352 + 385 \cos[t] + 284 \cos[2t] + 175 \cos[3t] + 144 \cos[4t] + 70 \cos[5t] + 60 \cos[6t]) \sin[t]$$

Eine andere Darstellung dieser Faltung zeigt die Behauptung
(FullSimplify, in der "cdf-Vorlage" direkt mit nur einem Mausklick zu machen)

In[13]:= `FullSimplify[Faltung[t]]`

Out[13]=
$$(1 + \cos[t]) \sin[t] + \frac{1}{3} \sin[3t] + \frac{1}{4} \sin[4t] + \frac{1}{5} \sin[5t] + \frac{1}{6} \sin[6t] + \frac{1}{7} \sin[7t]$$

Da $\cos[t] \sin[t] = \sin[2t]/2$ ist, ist also die Faltung genau wieder die
entsprechende Partialsumme der Fourierreihe des Sägezahns.

Berechnung der Fourier-Koeffizienten

Den k-ten Fourier-Koeffizienten der Fourierreihe in exponentieller Form
($k \neq 0$) erhält man mit dem folgenden Befehl, hier wieder am Beispiel des
Sägezahns, die Fourierreihen-Entwicklung in der Darstellung mit komplexen
Exponentialfunktionen mit dem Befehl FourierSeries

In[14]:= `FourierCoefficient[f[t], t, k]`

Out[14]=
$$-\frac{i}{2k}$$

In[15]:= `FourierSeries[f[t], t, 7]`

Out[15]=
$$\frac{1}{2} i e^{-it} - \frac{1}{2} i e^{it} + \frac{1}{4} i e^{-2it} - \frac{1}{4} i e^{2it} + \frac{1}{6} i e^{-3it} - \frac{1}{6} i e^{3it} + \frac{1}{8} i e^{-4it} - \frac{1}{8} i e^{4it} + \frac{1}{10} i e^{-5it} - \frac{1}{10} i e^{5it} + \frac{1}{12} i e^{-6it} - \frac{1}{12} i e^{6it} + \frac{1}{14} i e^{-7it} - \frac{1}{14} i e^{7it}$$

In[16]:= `FullSimplify[FourierSeries[f[t], t, 7]]`

Out[16]=
$$(1 + \cos[t]) \sin[t] + \frac{1}{3} \sin[3t] + \frac{1}{4} \sin[4t] + \frac{1}{5} \sin[5t] + \frac{1}{6} \sin[6t] + \frac{1}{7} \sin[7t]$$


3. Fejér-Kerne, Glättungen, Beseitigung des Gibbs-Phänomens

Die Fejér-Kerne sind arithmetische Mittel der Dirichlet-Kerne.
Wir definieren den 2π -periodischen Fejér-Kern $\text{Fej}[t, N]$ vom Grad $N-1 \geq 1$:

```
In[17]:= Fej[t_, N_] = 1 / N (1 + Sum[Dir[t, k], {k, 1, N - 1}])
```


$$\text{Out[17]} = \frac{1}{N} \left(1 + \frac{1}{2} \left(-\text{Csc}\left[\frac{t}{2}\right]^2 \text{Sin}\left[\frac{1}{2}(-\pi + 2t)\right] + \text{Csc}\left[\frac{t}{2}\right]^2 \text{Sin}\left[\frac{1}{2}(-\pi + 2Nt)\right] \right) \right)$$

```
In[18]:= g3 = Plot[Fej[t, 7], {t, -2 Pi, 2 Pi}, PlotStyle -> Directive[
  Red, Thickness[0.005]], PlotRange -> All]
```



Wir betrachten noch Dirichlet-Kern und Fejér-Kern mit gleichem Grad 6 der zugehörigen trigonometrischen Polynome zum Vergleich:

```
In[19]:= Show[g2, g3]
```



Auch die Fejér-Kerne haben die Mittelwerte Eins, sind aber im Unterschied zu den Dirichlet-Kernen nicht-negativ. Im Unterschied zu den Dirichlet-Kernen konvergieren sie für wachsende N in jedem abgeschlossenen Intervall, das keine Punkte der Form $2k\pi$ enthält, gleichmäßig gegen Null. Das trigonometrische Polynom mit Grad N für den Fejér-Kern $\text{Fej}[t, N+1]$ hat N Nullstellen pro Periode.

Hier der Mittelwert eines solchen Kerns:

```
In[20]:= Integrate[Fej[t, 5], {t, 0, 2 Pi}] / (2 Pi)
```

Out[20]= 1

Da es stetige periodische Funktionen gibt, deren Fourierreihen-Entwicklungen fast

nirgendwo überhaupt einen Grenzwert besitzen, war es 1904 ein bedeutendes Ergebnis von **L. Fejér**, **dass die arithmetischen Mittel der Partialsummen der Fourierreihe einer stetigen periodischen Funktion f punktweise, ja sogar gleichmäßig gegen f konvergieren.**


Überdies tritt bei der Approximation von Funktionen mit Sprungstellen das Gibbs-Phänomen nicht mehr auf. Durch die geringere Gewichtung hochfrequenter Anteile in der Approximation wird die Näherung geglättet und weniger wellig. Der Preis dafür ist ein größerer Fehler im quadratischen Mittel im Vergleich mit einer Näherung durch eine Partialsumme der Fourierreihe mit gleichem Grad. (vgl. [1] S. 25, S. 47 und S. 105-111)

Gleichbedeutend mit der Bildung solcher arithmetischer Mittel ist die periodische Faltung mit Fejér-Kernen. Wir zeigen dies am Beispiel einer Partialsumme des Sägezahns: Blau der Sägezahn, Rot die Näherung mit der Fejérschen Mittelung:

```
In[21]:= f1[t_, N_] := Sum[(1 - k / (N + 1)) Sin[k t] / k, {k, 1, N}]
```

```
In[22]:=
```


```
g4 := Plot[f1[t, 10], {t, -2 Pi, 2 Pi}, PlotStyle -> Directive[
  Red, Thickness[0.005]], PlotRange -> All]
g5 := Plot[Saegezahn[t + 2 Pi] + Saegezahn[t], {t, -2 Pi, 2 Pi},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
Show[g4, g5]
```



Zum Vergleich die Grafik der periodischen Faltung zwischen Sägezahn und unserem Fejér-Kern $\text{Fej}[t, 11]$:

(Die Berechnung kostet etwas Zeit, also Geduld; diese Berechnung ist natürlich aus praktischer Sicht zu umständlich, wenn man wie oben durch f_1 ja das resultierende trigonometrische Polynom direkt angeben kann.)


```
In[25]:= f2[t_, N_] := NIntegrate[Saegezahn[s] Fej[t - s, N], {s, 0, 2 Pi}] / (2 Pi)
g6 = Plot[f2[t, 11], {t, -2 Pi, 2 Pi}, PlotStyle -> Directive[
  Red, Thickness[0.005]], PlotRange -> All]
```



Zur Verdeutlichung (und wieder mit etwas Geduld bei der Berechnung durch *Mathematica*) die Differenz zwischen der direkt angegebenen Mittelbildung und der 2π -periodischen Faltung. Beide Sägezahn-Näherungen stimmen überein. Gezeigt werden Fehler aufgrund der numerischen Berechnung des Faltungsintegrals. Gewählt wurde ein niedriger Grad der trigonometrischen Näherungspolynome, um die Berechnungszeit zu verkürzen (Integrale mit stark oszillierenden Integranden sind ein numerisches Problem). Man beachte, dass nach obiger Definition wie in [1] der Fejér-Kern $\text{Fej}[t, N+1]$ zum trigonometrischen Näherungspolynom vom Grad N gehört, um Approximationen gleichen Grades zu erhalten.

Bis auf numerische Näherungsfehler stimmen beide Funktionen überein. Das ist so natürlich umständlich und zeitraubend, **da man ja weiß, dass es so ist**. Trotzdem hier mal so gemacht:

```
In[27]:= Plot[Abs[f1[t, 3] - f2[t, 4]], {t, 0, Pi}, PlotStyle -> Directive[
  Blue, Thickness[0.005]], PlotRange -> All]
```



Es gibt eine Vielzahl anderer sog. **Summationskerne**, anders gesagt Gewichtsfunktionen im Spektralbereich (vg. [1], S. 110-111 und Referenzen wie z.B. [2] A. Zygmund oder [3] E. Rietsch, Wietze und dort genannte weitere Quellen). Sie finden vielfach Verwendung in der Signalverarbeitung, wenn man punktweise gute Näherungen erzeugen will.

4. Eigenschaften von Fourierreihen, Rechnen mit Fourierreihen

Nachfolgend Beispiele dazu, wie sich die Fourierkoeffizienten verändern (oder auch nicht), wenn man eine periodische Funktion verschiebt, spiegelt, oder ihre Amplituden moduliert.

1. Für T-periodische Funktionen ist klar, dass das Integrationsintervall für die Berechnung der Fourierkoeffizienten verschoben werden kann und es dabei nur auf die Periodendauer ankommt. Auch klar ist, dass gerade periodische Funktionen eine Cos-Reihe besitzen, ungerade eine Sin-Reihe. Ebenso, dass bei Zeitumkehr von $f(t)$ zu $f(-t)$ die Fourierkoeffizienten c_k zu c_{-k} transformiert werden (Substitutionsregel für Integrale) und bei komplexer Konjugation von f zu \bar{c}_{-k} (vgl. [1], Kapitel 4)
2. Die zu einer periodischen Funktion $f(t)$ ähnliche Funktion $f(at)$, $a > 0$, hat die gleichen Fourierkoeffizienten. Allerdings ändert sich deren Frequenzzuordnung. Wir betrachten als Beispiel den Sägezahn $\text{Saegezahn}(t)$ und die skalierte ähnliche Funktion $f(t) = \text{Saegezahn}(2t)$, ihre Fourierreihen-Entwicklungen bis zu einem gewissen Grad und ihre Fourierkoeffizienten. Wir sehen die gleichen Fourierkoeffizienten/Amplituden, aber eben den Sinus-Schwingungen mit doppelten Frequenzen zugeordnet im Vergleich zum ursprünglichen Sägezahn. Anschaulich gesprochen: Das Signal f läuft doppelt so schnell ab, sieht ansonsten aber gleich, mathematisch "ähnlich" aus.


Daher genügt es, in Formelsammlungen die Fourierreihen auf eine feste Periode, etwa $T=2\pi$, bezogen anzugeben.

In[28]:=

```
FourierTrigSeries[f[2 t], t, 8]
```

Out[28]= $\text{Sin}[2 t] + \frac{1}{2} \text{Sin}[4 t] + \frac{1}{3} \text{Sin}[6 t] + \frac{1}{4} \text{Sin}[8 t]$

In[29]:= `Plot[Sin[2 t] + $\frac{1}{2}$ Sin[4 t] + $\frac{1}{3}$ Sin[6 t] + $\frac{1}{4}$ Sin[8 t],
{t, -13.194689145077131, 13.194689145077131},
PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]`



3. Translationen (Verschiebungen des Signals)


Es ist klar, dass eine "Zeitverschiebung" eines T-periodischen Signals nicht dessen Amplituden, wohl aber seine Phasen verändert, und zwar wird für $f(t + t_0)$ jeder Fourierkoeffizient mit $\text{Exp}(i k \omega_0 t_0)$ zur Phasenverschiebung multipliziert ($\omega_0 = 2\pi/T$) (vgl. [1], S. 35). Wir verschieben zur Veranschaulichung den 2π -periodischen Sägezahn um $t_0 = \pi$ und betrachten die Fourierkoeffizienten:

Sie sind dann im Vergleich mit der unverschobenen Funktion jeweils $(-1)^k$ multipliziert, wodurch eine alternierende Reihe entsteht.

```
In[30]:= ftransl[t_] = Saegezahn[t + Pi]
```

```
Out[30]:=  $-\frac{1}{2} t (-\text{UnitStep}[-\pi + t] + \text{UnitStep}[\pi + t])$ 
```

```
In[31]:= Plot[- $\frac{1}{2} t (-\text{UnitStep}[-\pi + t] + \text{UnitStep}[\pi + t])$ , {t, -Pi, Pi},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
```




Hier die Fourierreihen-Entwicklung bis zum Grad 5 und ihr Plot:

```
In[32]:= trigpol5[t_] = FourierTrigSeries[ftransl[t], t, 5]
```

```
Out[32]:=  $-\text{Sin}[t] + \frac{1}{2} \text{Sin}[2 t] - \frac{1}{3} \text{Sin}[3 t] + \frac{1}{4} \text{Sin}[4 t] - \frac{1}{5} \text{Sin}[5 t]$ 
```

```
In[33]:= Plot[trigpol5[t], {t, -2 Pi, 2 Pi},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
```



4. Amplitudenmodulation, Translation des Spektrums

Amplitudenmodulation bewirkt eine Verschiebung des Spektrums (vgl. [1] S. 35-36). Wir betrachten das zuletzt erzeugte trigonometrische Polynom und multiplizieren mit $\text{Cos}[5 t]$. Es entsteht eine spektrale Verschiebung nach links und nach rechts, d.h. zwei Seitenbänder und eine Halbierung der Spektralwerte.

Amplitudenmodulation ist daher in der Nachrichtentechnik eine Methode, ein Signal mit beschränkter Bandbreite zur Übertragung in ein gewünschtes Frequenzband zu verschieben und zum Empfang mit der gleichen Amplitudenmodulation in das ursprüngliche Frequenzband zurückzuholen. Seitenbänder werden dabei jeweils unterdrückt (einmal das linke, einmal das rechte), man hat also bei diesem Verfahren ganz erhebliche Leistungsverluste in Kauf zu nehmen.


Ein Anwendungsbeispiel kennt wohl jeder an Rundfunkempfängern mit Sendern, die über AM (Amplitudenmodulation) zu empfangen sind.

```
In[34]:= ampmod[t_] = trigpol5[t] Cos[5 t]
```

$$\text{Out[34]} = \text{Cos}[5 t] \left(-\text{Sin}[t] + \frac{1}{2} \text{Sin}[2 t] - \frac{1}{3} \text{Sin}[3 t] + \frac{1}{4} \text{Sin}[4 t] - \frac{1}{5} \text{Sin}[5 t] \right)$$

Im nachfolgenden Plot ist die Amplitudenmodulation deutlich zu erkennen :

```
In[35]:= Plot[ampmod[t], {t, -2 Pi, 2 Pi},
  PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
```



Nun die Spektren zum Vergleich: Zuerst das Ausgangsspektrum

In[36]:= **FourierCoefficient**[trigpol5[t], t, k]

$$\text{Out[36]=} \begin{cases} -\frac{i}{10} & k == -5 \\ \frac{i}{10} & k == 5 \\ -\frac{i}{8} & k == 4 \\ \frac{i}{8} & k == -4 \\ -\frac{i}{6} & k == -3 \\ \frac{i}{6} & k == 3 \\ -\frac{i}{4} & k == 2 \\ \frac{i}{4} & k == -2 \\ -\frac{i}{2} & k == -1 \\ \frac{i}{2} & k == 1 \\ 0 & \text{True} \end{cases}$$

Hier das verschobene Spektrum des amplitudenmodulierten Signals und seine Fourierreihen-Darstellung:

In[37]:= **FourierCoefficient**[ampmod[t], t, k]

$$\text{Out[37]=} \begin{cases} -\frac{i}{20} & k == -10 \\ \frac{i}{20} & k == 10 \\ -\frac{i}{16} & k == -1 \quad || \quad k == 9 \\ \frac{i}{16} & k == -9 \quad || \quad k == 1 \\ -\frac{i}{12} & k == -8 \quad || \quad k == 2 \\ \frac{i}{12} & k == -2 \quad || \quad k == 8 \\ -\frac{i}{8} & k == -3 \quad || \quad k == 7 \\ \frac{i}{8} & k == -7 \quad || \quad k == 3 \\ -\frac{i}{4} & k == -6 \quad || \quad k == 4 \\ \frac{i}{4} & k == -4 \quad || \quad k == 6 \\ 0 & \text{True} \end{cases}$$


In[38]:= **FourierTrigSeries**[ampmod[t], t, 10]

$$\text{Out[38]=} -\frac{\text{Sin}[t]}{8} + \frac{1}{6} \text{Sin}[2 t] - \frac{1}{4} \text{Sin}[3 t] + \frac{1}{2} \text{Sin}[4 t] - \frac{1}{2} \text{Sin}[6 t] + \frac{1}{4} \text{Sin}[7 t] - \frac{1}{6} \text{Sin}[8 t] + \frac{1}{8} \text{Sin}[9 t] - \frac{1}{10} \text{Sin}[10 t]$$

Zur Veranschaulichung sondern wir das "obere Seitenbandsignal" im Frequenzband von 6 Hz bis 10 Hz aus, unterwerfen es erneuter Amplitudenmodulation mit $\text{Cos}[5 t]$, filtern (per Hand) im Ergebnis dann das untere Seitenband bis 5 Hz wieder heraus und plotten das Ergebnis. Man sieht wieder die Form der Funktion $\text{trigpol5}[t]$, nun aber mit erheblich verkleinerten Amplituden.

```
In[39]:= einseitenbandsignal[t_] :=
  -  $\frac{1}{2}$  Sin[6 t] +  $\frac{1}{4}$  Sin[7 t] -  $\frac{1}{6}$  Sin[8 t] +  $\frac{1}{8}$  Sin[9 t] -  $\frac{1}{10}$  Sin[10 t]
```

```
In[40]:= Plot[einseitenbandsignal[t], {t, -2 Pi, 2 Pi},
  PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]
```



Jetzt also erneute Amplitudenmodulation und danach "herausgenommen" nur das untere Seitenband:


```
In[41]:= TrigReduce[Cos[5 t] einseitenbandsignal[t]]
```

Out[41]= $\frac{1}{240} (-60 \sin[t] + 30 \sin[2 t] - 20 \sin[3 t] + 15 \sin[4 t] - 12 \sin[5 t] - 60 \sin[11 t] + 30 \sin[12 t] - 20 \sin[13 t] + 15 \sin[14 t] - 12 \sin[15 t])$

```
In[42]:= zurueckmoduliert[t_] :=
   $\frac{1}{240} (-60 \sin[t] + 30 \sin[2 t] - 20 \sin[3 t] + 15 \sin[4 t] - 12 \sin[5 t])$ 
```

Im Ergebnis wieder trigpol5[t] mit stark erniedrigten Amplituden:

```
In[43]:= Plot[zurueckmoduliert[t], {t, -2 Pi, 2 Pi},
  PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]
```



5. Glattheitseigenschaften und Abfallverhalten des Spektrums

Je glatter eine periodische Funktion ist, desto schneller gehen für $|k| \rightarrow \infty$ die Spektralwerte gegen Null. (vgl. [1] S. 39-43).

Wir betrachten zwei Beispiele. Das erste ist die Funktion $g(t) = t^2$, 2π -periodisch fortgesetzt. Sie ist stetig auf \mathbb{R} , aber nicht stetig differenzierbar. Ihr Spektrum fällt betragsmäßig ab wie $1/k^2$. Das zweite Beispiel ist die Funktion $f(t) = t(\pi+t)$ auf $[-\pi, 0]$ und $f(t) = t(\pi-t)$ auf $[0, \pi]$. Ihre 2π -periodische Fortsetzung ist stetig differenzierbar mit stückweise stetiger zweiter Ableitung. Ihr Spektrum fällt für $|k| \rightarrow \infty$ ab wie $1/|k|^3$.

```
In[44]:= g[t_] := t^2
FourierTrigSeries[g[t], t, 6]
```

$$\text{Out[45]} = \frac{\pi^2}{3} + 4 \left(-\cos[t] + \frac{1}{4} \cos[2t] - \frac{1}{9} \cos[3t] + \frac{1}{16} \cos[4t] - \frac{1}{25} \cos[5t] + \frac{1}{36} \cos[6t] \right)$$


```
In[46]:= f[t_] := Piecewise[{{t (Pi + t), t <= 0}, {t (Pi - t), t >= 0}}]
FourierTrigSeries[f[t], t, 10]
```

$$\text{Out[47]} = \frac{8 \sin[t]}{\pi} + \frac{8 \sin[3t]}{27\pi} + \frac{8 \sin[5t]}{125\pi} + \frac{8 \sin[7t]}{343\pi} + \frac{8 \sin[9t]}{729\pi}$$

Nachfolgend ist zu sehen, dass f eine stetig differenzierbare 2π -periodische Fortsetzung hat, welche aber nur eine stückweise stetige zweite Ableitung besitzt.

```
In[48]:= f'[t]
Plot[f'[t], {t, -Pi, Pi},
PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
```

$$\text{Out[48]} = \begin{cases} \pi + 2t & t < 0 \\ \pi & t == 0 \\ \pi - 2t & \text{True} \end{cases}$$



Die Zusammenhänge von (lokalen) Differenzierbarkeitseigenschaften einerseits und dem (globalen) Abfallverhalten des Spektrums andererseits zeigen, dass etwa kleine lokale Störungen bei Verlust von Differenzierbarkeitseigenschaften das gesamte Spektrum einer Funktion drastisch verändern können. In der Praxis der Signalverarbeitung sind solche Störungen oft unvermeidbar und ergeben zwangsläufig große Herausforderungen bei Problemen, in denen aus Daten, welche Funktionsverläufe beschreiben, Schätzungen

über die Spektren der beobachteten Signale gewonnen werden sollen. Als mathematische Hilfsmittel kommen dabei vorzugsweise Varianten von Tiefpassfiltern vor der Spektralschätzung für das sog. "Denoising" zum Einsatz. Umgekehrt benötigt man zur Nachbildung von steilen Signalfanken durch ein trigonometrisches Polynom hochfrequente Anteile in der Näherung, m.a.W. gesagt eine große Bandbreite.

6. Fourierreihen von Ableitungen und Integralfunktionen

Die Fourierkoeffizienten der Ableitung einer T-periodischen Funktion f mit Fourierkoeffizienten c_k sind die Koeffizienten $ik\omega_0 c_k$ mit $\omega_0=2\pi/T$. Es kann natürlich sein, dass die entstehende Reihe an keiner einzigen Stelle mehr konvergiert, wie das Beispiel der Sägezahnreihe sofort lehrt. Ist die Ableitung von f stückweise stetig und hat die Reihe mit den Koeffizienten $ik\omega_0 c_k$ an einer Stelle t_0 aber überhaupt einen Grenzwert, dann ist dieser Grenzwert gegeben durch $(f'(t_0+) + f'(t_0-))/2$ nach dem Satz von Fejér (vgl. [1], S. 107).


Einfacher ist es bei der **Integration von Fourierreihen f** : Sie dürfen immer gliedweise integriert werden.

Als Integralfunktion $\int_0^t f(x) dx$ ergibt sich eine periodische Funktion, die um die Rampe $c_0 t + F_0$ schwingt, wobei c_0 der Mittelwert von f ist und F_0 der Mittelwert der Integralfunktion $\int_0^t (f(x) - c_0) dx$. Wir betrachten als Beispiel eine Partialsumme der Fourierreihe des um 1/2 angehobenen Sägezahns:

```
In[50]:= f[t_] = 1/2 + Sum[Sin[k t] / k, {k, 1, 4}]
```

```
Out[50]= 1/2 + Sin[t] + 1/2 Sin[2 t] + 1/3 Sin[3 t] + 1/4 Sin[4 t]
```

```
In[51]:= Plot[f[t], {t, -2 Pi, 2 Pi},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
```



Hier die Integralfunktion:

```
In[52]:= ff[t_] = Integrate[f[x], {x, 0, t}]
```

```
Out[52]= 1 + t/2 - Cos[t] + Sin[t]^2/2 + 2/9 Sin[3 t/2]^2 + 1/8 Sin[2 t]^2
```


```
In[53]:= TrigReduce[1 + t/2 - Cos[t] + Sin[t]^2/2 + 2/9 Sin[3t/2]^2 + 1/8 Sin[2t]^2]
```

```
Out[53]:= 1/144 (205 + 72 t - 144 Cos[t] - 36 Cos[2 t] - 16 Cos[3 t] - 9 Cos[4 t])
```

Das Ergebnis ist also eine periodische Funktion, die um die Rampe $c_0 t + F_0$ schwingt, c_0 und F_0 wie oben angegeben. Die Funktion geht bei Null durch den Nullpunkt.

Übergang von der Partialsumme zu den kompletten Fourierreihen des Sägezahns und seiner Integralfunktion zeigen im ersten Fall einen Abfall des Spektrums wie $1/k$ (der Sägezahn ist nicht stetig), bei der Integralfunktion einen Abfall wie $1/k^2$ (die Integralfunktion ist stetig, aber nicht stetig differenzierbar). Man vgl. nochmal weiter oben.

```
In[54]:= Plot[ff[t], {t, -4 Pi, 16 Pi},
  PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
```



```
In[55]:= F0 = Integrate[Integrate[f[x] - 1/2, {x, 0, t}], {t, 0, 2 Pi}] / (2 Pi)
```

```
Out[55]:= 205/144
```

7. Zu T-periodischen Fourierreihen-Entwicklungen als Orthogonalprojektionen im Vektorraum $L^2([0, T])$ und zur Verallgemeinerung dieses Aspekts.

Wir kommen **zum wahrscheinlich wichtigsten Aspekt**, nämlich dass Fourierreihen-Entwicklungen T-periodischer Funktionen/Signale f als Orthogonalprojektionen von f auf endlich-dimensionale Unterräume des Raums $L^2([0, T])$ der quadratisch-integrierbaren Funktionen $f: [0, T] \rightarrow \mathbb{C}$ zu lesen sind.

Eine Partialsumme vom Grad $\leq N$ der Fourierreihe einer T-periodischen quadratisch integrierbaren Funktion f ist die **beste Approximation** für f im endlich-dimensionalen Teilraum von $L^2([0, T])$, der durch die trigonometrischen Funktionen $1, \cos[2\pi kt/T], \sin[2\pi kt/T], k=1..N$, erzeugt wird (vgl. [1], S. 47-48). Sie minimiert die Norm des Fehlers (und damit die Leistungsverluste bei der Approximation). Sie ist daher die **Orthogonalprojektion der Funktion f auf den so erzeugten Teilraum**. Das innere Produkt, mit dem Orthogonalität "gemessen" wird, ist dabei das in $L^2([0, T])$ üblicherweise betrachtete (vgl. [1], S. 9). Diese Optimierung bezüglich des Fehlers im quadratischen Mittel ist einer der Hauptgründe für die vielfache Verwendung von

Fourier-Entwicklungen in der Technik, wo es oft weniger auf gute punktweise Näherungen ankommt als auf Mittelwerte wie sie etwa durch den Begriff der Leistung definiert sind. Die Reihe der Betragsquadrate der (komplexen) Fourierkoeffizienten konvergiert gegen die Gesamtleistung des "Signals" f (Parsevalgleichung, siehe [1], S. 48 und S. 111).

Als Beispiel hier wieder der Sägezahn: Zuerst die Reihe der Betragsquadrate der (komplexen) Fourierkoeffizienten (also $1/(2k^2)$ summieren, nicht $1/k^2$ für $k=1..∞$), dann die Leistung über das Integral berechnet:

```
In[56]:= Sum[1 / (2 k ^ 2) , {k, 1, Infinity}]
```

```
Out[56]=  $\frac{\pi^2}{12}$ 
```

```
In[57]:= Integrate[(Pi - t)^2/4, {t, 0, 2 Pi}] / (2 Pi)
```

```
Out[57]=  $\frac{\pi^2}{12}$ 
```

Die Wurzel ist die L^2 -Norm der Funktion, die in der Technik auch als **Effektivwert** (etwa, wenn f eine Wechselspannung beschreibt) bezeichnet wird.

Ein weitreichendes neues Konzept :

Orthogonalprojektionen als optimale Näherungen von Signalen in endlich-dimensionalen Teilräumen von $L^2([a, b])$

Fourierreihen-Entwicklungen von Funktionen aus $L^2([0, T])$ mit dem Orthogonalsystem der trigonometrischen Funktionen $1, \cos[2\pi kt/T], \sin[2\pi kt/T], k=1..N$, sind nur ein erstes Beispiel für das allgemeine Konzept, Funktionen aus dem Raum $L^2([a, b])$ - auch für andere Intervalle $[a, b]$ - mit Hilfe eines vollständigen Orthogonalsystems als Funktionenreihen darzustellen. Die Partialsummen solcher Reihen sind dann wieder Orthogonalprojektionen in die von den beteiligten Funktionen erzeugten endlich-dimensionalen Teilräume von $L^2([a, b])$ und damit die besten Approximationen an die entwickelte Funktion in den jeweiligen Teilräumen im Sinne der Norm von $L^2([a, b])$. Es gibt viele solche Orthogonalsysteme von Funktionen in $L^2([a, b])$, die in der Mathematik und der Technik verwendet werden.

Wir betrachten zunächst ein einziges Beispiel, weitere dann in nachfolgenden Notebooks.

Als Beispiel möge die Aufgabe dienen, die Funktion $f[t]=\sin[3t]$ im Intervall $[-1, 1]$ durch ein Polynom anzunähern. Polynome sind aus vielfachen Gründen häufig bevorzugte Näherungsfunktionen (sie gestatten eine besonders einfache Verarbeitung wie differenzieren, integrieren, Werte berechnen usw.). Als erste Methode einer Näherung lernt man gewöhnlich im ersten Semester die **Taylor-Entwicklung** kennen. Dabei lernt man auch, dass ein Taylorpolynom zwar am Entwicklungspunkt nach seiner Konstruktion den exakten Funktionswert liefert, jedoch wird der Fehler mit der Entfernung vom Entwicklungspunkt oft immer größer.

Eine **polynomiale Näherung im quadratischen Mittel** hat dagegen eher die Eigenschaft,


dass die Näherungsfunktion meist um die "Zielfunktion" herumschwingt und über das ganze Näherungsintervall in ihrer Nähe bleibt. Wir vergleichen zwei solche Näherungen an eine Sin-Funktion, erstens mit dem Taylorpolynom bis zum Grad 5, zweitens eine Näherung im quadratischen Mittel mit dem **Orthogonalsystem der Legendre-Polynome im reellen Vektorraum $L^2([-1, 1])$** . Als inneres Produkt wählen wir dabei $\langle f, g \rangle = \int_{-1}^1 f(x) g(x) dx$ (diesmal ohne Vorfaktor beim Integral wie vorher).

Zunächst das Taylorpolynom vom Grad 5 und ein Plot der Sin-Funktion (blau) und dieses Näherungspolynoms (rot):

```
In[58]:= n1[x_] = Normal[Series[Sin[3 x], {x, 0, 5}]]
```

$$\text{Out[58]} = 3x - \frac{9x^3}{2} + \frac{81x^5}{40}$$

```
In[59]:= p1 := Plot[Sin[3 x], {x, -1, 1},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
p2 := Plot[n1[x], {x, -1, 1}, PlotStyle -> Directive[Red, Thickness[0.005]],
  PlotRange -> All]
Show[
  p1,
  p2]
```



Man erkennt ein typisches Anwachsen des Fehlers mit der Entfernung zum Entwicklungspunkt Null.

Nun zur Approximation mit dem Orthogonalsystem der Legendre-Polynome. *Mathematica* kennt diese Polynome als LegendreP. Wir sehen uns diese Polynome bis zum Grad 5 an. Sie sind in normierter Form wie in [1], S. 363 definiert durch

```
In[62]:= LPol[k_, x_] = 1 / (2^k k!) D[(x^2 - 1)^k, {x, k}] / Sqrt[2 / (2k + 1)]
```

$$\text{Out[62]} = \frac{2^{-\frac{1}{2}-k} \partial_{\{x, k\}} (-1 + x^2)^k}{\sqrt{\frac{1}{1+2k} k!}}$$

Hier als Beispiel das normierte Legendre-Polynom vom Grad 3 und seine L^2 -Norm mit dem oben gewählten inneren Produkt

In[63]:=

Expand[LPol[3, x]]

$$\text{Out[63]} = -\frac{3}{2} \sqrt{\frac{7}{2}} x + \frac{5}{2} \sqrt{\frac{7}{2}} x^3$$

In[64]:= **Integrate**[LPol[3, x]^2, {x, -1, 1}]

Out[64]= 1

Mathematica kennt natürlich die Legendre-Polynome als LegendreP[k,x], sie sind dort aber anders normiert. Hier zum Vergleich LegendreP[3,x] in *Mathematica* und seine Norm mit unserem inneren Produkt.

In[65]:=

Expand[LegendreP[3, x]]


$$\text{Out[65]} = -\frac{3x}{2} + \frac{5x^3}{2}$$

In[66]:= **Sqrt**[**Integrate**[LegendreP[3, x]^2, {x, -1, 1}]]

$$\text{Out[66]} = \sqrt{\frac{2}{7}}$$

Plot der ersten 6 wie oben normierten Legendre-Polynome. Beachten Sie die Symmetrieeigenschaften und die Anzahl der Nullstellen dieser Polynome in [-1,1]:

In[67]:= **Plot**[**Evaluate**[**Table**[LegendreP[n, x] / Sqrt[2 / (2 n + 1)], {n, 0, 5}], {x, -1, 1},
PlotStyle → **Directive**[Red, Thickness[0.005]], **PlotRange** → All]



Nun zur Approximation von Sin[3x] im quadratischen Mittel mit den Legendre-Polynomen bis zum Grad 5:


In[68]:= **Faktor**[n_] := **NIntegrate**[LegendreP[n, x] Sin[3 x], {x, -1, 1}]
n2[x_] = **Expand**[**Sum**[Faktor[n] LegendreP[n, x] / (2 / (2 n + 1)), {n, 0, 5}]]

Out[69]= 0. + 2.97177 x - 4.23916 x³ + 1.42043 x⁵

Hier nun das so erhaltene Näherungspolynom vom Grad 5 wie oben zur Ansicht rot

mit einem Offset von 0.02 gezeichnet, damit man die Näherung überhaupt von der Zielfunktion $\sin[3x]$ unterscheiden kann. Dass sie über das ganze Intervall $[-1,1]$ betrachtet für viele Zwecke besser als das Taylorpolynom ist, sehen wir anschließend noch einmal ganz deutlich, indem wir den absoluten Fehlerverlauf des Taylorpolynoms in Blau und den der Näherung im quadratischen Mittel mit den Legendre-Polynomen in Rot betrachten.


```
In[70]:= p3 := Plot[n2[x] + 0.02, {x, -1, 1},
  PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
Show[p1, p3]
```



```
In[72]:= p4 := Plot[Abs[Sin[3 x] - n1[x]], {x, -1, 1},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
p5 := Plot[Abs[Sin[3 x] - n2[x]], {x, -1, 1},
  PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
```

Nun die Fehlerkurven :

```
In[74]:= Show[p4, p5]
```




Sie erkennen, dass das Konzept der Entwicklung nach einem Orthogonalsystem von Funktionen in Räumen mit einem inneren Produkt ein mächtiges Werkzeug sein kann, wenn man komplizierte Funktionen durch Näherungen mit einfacher zu handhabenden Funktionen ersetzen will oder muss. Zum Abschluss dieses ersten Notebooks sei noch darauf hingewiesen, dass auch bei der Verwendung anderer Orthogonalsysteme als den trigonometrischen Funktionen wie in der klassischen Fourier-Analyse bei Funktionen mit Sprungstellen vielfach wieder das Gibbs'sche Phänomen auftritt. Zur Bekämpfung kann man auch hier Faltungen mit geeigneten

Summationskernen verwenden, wie wir es mit den Fejér-Kernen weiter oben bei Fourierreihen kennen gelernt haben.

Als Beispiel nehmen wir die Signum-Funktion im Intervall $[-1,1]$ und verwenden zur Näherung wie zuletzt die Legendre-Polynome, diesmal bis zum Grad 25. Der Versuch einer guten Approximation mit einem Taylorpolynom über das gesamte Intervall wäre wegen der Sprungstelle schon a priori ein denkbar ungeeigneter Versuch. Aus Symmetriegründen müssen wir nur die ungeraden Polynome berücksichtigen. Damit die numerisch berechneten Werte der Integrale für die inneren Produkte konvergieren, wird nachfolgend die "WorkingPrecision" von *Mathematica* hochgesetzt.

In[75]=

```
f[x_] := 2 UnitStep[x] - 1
p6 = Plot[f[x], {x, -1, 1},
  PlotStyle -> Directive[Blue, Thickness[0.005]], PlotRange -> All]
```




In[77]=

```
Faktor2[n_] := NIntegrate[LegendreP[n, x] f[x], {x, -1, 1}, WorkingPrecision -> 40]
n3[x_] = Expand[Sum[Faktor2[n] LegendreP[n, x] / (2 / (2 n + 1)), {n, 1, 25, 2}]]
```

Out[78]=

```
16.86141919703842972921847831457853317261 x -
651.9748756188192828631144948303699493408 x3 +
13 339.40595516104252737932256422936916351 x5 -
149 728.0260273178242869107634760439395905 x7 +
1.018982399352579637508142695878632366657 × 106 x9 -
4.486891094008714304663953953422605991364 × 106 x11 +
1.328810054764119236381247901590541005135 × 107 x13 -
2.698117177863906868537924310658127069473 × 107 x15 +
3.763593381752205731416438538872171193361 × 107 x17 -
3.544658586691828112691382557386532425880 × 107 x19 +
2.153319808105988098390071172616444528103 × 107 x21 -
7.615624961474692398155639239121228456497 × 106 x23 +
1.191083743974641891071541976998560130596 × 106 x25
```

```
In[79]:= p7 := Plot[n3[x], {x, -1, 1},
  PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
Show[p6, p7]
```



Wir sehen das Gibbsche Phänomen und wir sehen auch zunehmende Fehler an den Rändern des Intervalls $[-1, 1]$. Das Verhalten am Rand kann man beeinflussen, wenn man beim inneren Produkt Gewichtsfunktionen einführt. Wir werden das in einem nachfolgenden Notebook in Beispielen zur Approximation und Interpolation durch Tschebyscheff-Polynome (oft auch Chebyshev-Polynome geschrieben) noch näher diskutieren und demonstrieren.

In nachfolgenden Notebooks werden wir weitere Orthogonalsysteme, insbesondere die gerade genannten Tschebyscheff-Polynome in konkreten Anwendungsbeispielen kennen lernen, die eine große Rolle in der Approximationstheorie, aber auch in Anwendungen der Elektrotechnik beim Schaltungsentwurf und in der Signalverarbeitung spielen.

Da in [1] (wie in anderen einführenden Lehrbüchern auch) vornehmlich Fourierreihen-Darstellungen stückweise stetig differenzierbarer periodischer Funktionen (vgl. etwa [4]) behandelt werden, soll hier zum Abschluss auch ein Beispiel einer Fourierreihe für eine periodische Funktion gezeigt werden, welche diese Voraussetzung nicht erfüllt. Es ist das Beispiel aus der Übungsaufgabe A6 in [1], Kapitel 6.

8. Die Fourierreihe für eine Funktion, die nicht stückweise stetig differenzierbar ist

In[81]=


```
f[t_] = Log[Abs[2 Sin[t / 2]]] (UnitStep[t] - UnitStep[t - 2 Pi])
```

Out[81]= $\text{Log}\left[2 \text{Abs}\left[\text{Sin}\left[\frac{t}{2}\right]\right]\right] (\text{UnitStep}[t] - \text{UnitStep}[-2 \pi + t])$

Nachfolgend der Plot der 2π -periodischen Fortsetzung im Intervall $[-2\pi, 2\pi]$. Da für $t \rightarrow 0$ oder $t \rightarrow 2\pi$ keine Grenzwerte existieren, ist die Funktion nicht stückweise stetig differenzierbar, wohl aber auf $[0, 2\pi]$ integrierbar. Sie besitzt für $t \neq 2k\pi$, k in \mathbb{Z} , eine Fourierreihen-Darstellung. Wir lassen von *Mathematica* eine Partialsumme dieser Fourierreihe berechnen (vgl. [1] S. 120). Obgleich die Reihe eine große Ähnlichkeit mit der Sägezahnreihe hat (dort stand $\sin(kt)$ in den Summanden der Partialsummen, hier steht stattdessen $-\cos(kt)$), ist die dargestellte Funktion

gänzlich verschieden vom Sägezahn. Auch im nachfolgenden Beispiel sehen Sie, dass es durchaus beachtenswerte Unterschiede zwischen Sinus-Reihen und Cosinus-Reihen gibt.

```
In[82]:= Plot[f[t + 2 Pi] + f[t], {t, -2 Pi, 2 Pi},
  PlotStyle -> Directive[Blue, Thickness[0.005]]]
```



```
In[83]:= FourierTrigSeries[f[t + 2 Pi] + f[t], t, 6]
```

Out[83]=
$$-\cos[t] - \frac{1}{2} \cos[2t] - \frac{1}{3} \cos[3t] - \frac{1}{4} \cos[4t] - \frac{1}{5} \cos[5t] - \frac{1}{6} \cos[6t]$$

9. Ein Beispiel einer zwar konvergenten trigonometrischen Reihe, die aber nicht die klassische Fourierreihe der durch die Reihe dargestellten Funktion sein kann, und ein Beispiel einer einfachen periodischen Funktion, die keine klassische Fourierreihen-Darstellung besitzt.

Das erste Beispiel wird behandelt in der Aufgabe A11 in [1], S. 46.

Im zweiten Beispiel betrachten wir eine der ersten periodischen Funktionen, an die man denken kann, nämlich die **Tangens-Funktion**. Sie ist unbeschränkt, also nicht stückweise stetig differenzierbar. Sie hat keine klassische Fourierreihen-Darstellung. Erst mit der Distributionentheorie gelingt es, auch für diese Funktion, die dann in einem gewissen Sinn als periodische Distribution aufgefasst werden kann, eine sogenannte verallgemeinerte Fourierreihen-Darstellung anzugeben (vgl. [1], S. 173).

```
In[84]:= f[t_] = Sum[Sin[k t] / Log[k], {k, 2, Infinity}]
```

Out[84]=
$$\sum_{k=2}^{\infty} \frac{\sin[kt]}{\log[k]}$$

Mathematica gibt die Definition zurück und kann keine explizite einfachere Darstellung der zugehörigen Funktion bestimmen. Man kann zeigen, dass die Reihe überall konvergiert,

in jedem Intervall $[h, 2\pi-h]$, $h>0$, sogar gleichmäßig gegen eine stetige Funktion. **Sie kann keine klassische Fourierreihe einer auf $[0, 2\pi]$ integrierbaren Funktion sein**, da die Reihe $\sum_{k=2}^{\infty} 1/(k \text{Log}(k))$ divergiert. Für die Sinus-Koeffizienten b_k der Fourierreihe einer integrierbaren Funktion muss aber die Reihe $\sum_{k=1}^{\infty} b_k/k$ konvergent sein (vgl. [1], S. 46, A11). Für eine Cos-Reihe gilt eine entsprechende Koeffizientenbedingung wiederum nicht. So ist etwa die Reihe $\sum_{k=2}^{\infty} \cos(kt)/\text{Log}(k)$ **in der Tat eine klassische Fourierreihe** (zu finden etwa in [2] **A. Zygmund, dort Kapitel V, Abschnitt 1**).

Schon diese Beispiele zeigen subtile mathematische Besonderheiten, sobald man sich für punktweise Darstellungen periodischer Funktionen durch Fourierreihen interessiert. Die Schwierigkeiten sind insbesondere durch den verwendeten Integralbegriff gegeben, denn Fourierkoeffizienten sind ja durch Integration zu berechnen.

Sie waren Anlass schon für die Entwicklung der Mengenlehre durch G. Cantor, die Entwicklung der Integralbegriffe von B. Riemann und H. Lebesgue, für zahlreiche Arbeiten großer Mathematiker wie G. H. Hardy, A. Zygmund u.v.a.m., wie auch schon für die Entwicklung des modernen Funktionsbegriffs durch A.L. Cauchy, P.L. Dirichlet und damit der gesamten mathematischen Entwicklung in der Analysis seit etwa Mitte des 19. Jahrhunderts.

```
In[85]= Sum[1 / (k Log[k]), {k, 2, Infinity}]
```

Sum::div : Sum does not converge. >>

$$\text{Out[85]= } \sum_{k=2}^{\infty} \frac{1}{k \text{Log}[k]}$$

Nun zum Tangens. Er hat keine klassische Fourierreihen-Darstellung.

Mathematica gibt kein Ergebnis für die Fourierreihen-Entwicklung zurück.

Die Darstellung des Tangens als (regularisierte) periodische Distribution behandeln wir in einem späteren Notebook über "Distributionen und Anwendungsbeispiele".

```
In[86]= FourierTrigSeries[Tan[t], t, 5]
```

```
Out[86]= FourierTrigSeries[Tan[t], t, 5]
```

Weitere empfehlenswerte Referenzen:

- [2] A. Zygmund Trigonometric Series
Cambridge University Press, Cambridge, 2003
- [3] E. Rietsch, Wietze A Fourier Series Kernel Based on Chebyshev Polynomials
Computing 18 (1977), 37 - 50
- [4] G. B. Folland Fourier Analysis And Applications
Wadsworth, Pacific Grove, 1992

u.v.a.m., die Sie im Literaturverzeichnis am Ende von [1] finden können.